/*
Howard Schuman	
Proposal title: Do Collective Memory and Autobiographical Memory Refer
to Different Developmental Stages?


HYPOTHESES

Stated-Hyp1: the "ages for memory of different types of events are much the same or basically different" (p. 3)

	Test-Hyp1: The average age for memory of personal events (AM) will differ from the average age for memory of national/global events (CM)

Stated-Hyp2: There will be no order effects related to whether the AM or CM question is asked first (p. 4)

	Test-Hyp2: There will be no difference in the average age of AM if the AM question is asked first or later.

	Test-Hyp3: There will be no difference in the average age of CM if the CM question is asked first or later.

********************************************************************************
NOTES
- each respondent got 5 am and 5 cm questions.

- We averaged the age of personal and national/word events within respondents 
(the alternative is to reshape the file such that each response is a single row, 
but that could be misleading because of heterogeneity in word effects).	

- For hypothesis 1, positive difference in the outcome measure means age of 
personal memory is older than that of collective memory. We are not using 
absolute difference because the direction of difference matters for the 
hypothesis test.
*/

clear all
use "Schuman601.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATIONS

* question order
	tab xtess170, nol
	clonevar questionorder=xtess170 
	
* OUTCOME MEASURES

* age when event occured
	forval i=1/10 {
	replace Q`i'A=. if Q`i'A==-1
	tab Q`i'A
	}
 	
	// 1-5 are personal events and 6-10 are national events
	
	* average age of personal events
	local personal "Q1A Q2A Q3A Q4A Q5A"
	egen personalage= rmean(`personal')
	tab personalage

	* average age of national/global events
	local national "Q6A Q7A Q8A Q9A Q10A"
	egen nationalage= rmean(`national')
	tab nationalage	
	
********************************************************************************

* ANALYSIS
	
*Test-Hyp1: The average age for memory of personal events (AM) will differ from the average age for memory of national/global events (CM)

	ttest personalage == nationalage
	* do not reject. 0.000 [age of personal memory is lower on average]
	tess, ttest init(Schuman601)
	
*Test-Hyp2: There will be no difference in the average age of AM if the AM question is asked first or later.
	reg personalage i.questionorder
	// reject. 0.000
	tess 2.questionorder, bonf(2)
	
*Test-Hyp3: There will be no difference in the average age of CM if the CM question is asked first or later.
	reg nationalage i.questionorder
	// reject. 0.000
	tess 2.questionorder, bonf(2)
